
Introduction to Cypher



We will learn how to

• represent nodes and relationships in Cypher

• create and find nodes

• add properties

• use constraints and create nodes uniquely

• create relationships

• find patterns

• delete data



Setup



1. Start the server.

2. It should be running on: http://localhost:7474 

3. Log-in with default credentials 

○ user: neo4j password: neo4j

4. Choose a new password

We’re good to go!

Make sure you’ve got Neo4j running

http://localhost:7474


:play movies

Type the following command into the query editor:

:play movies



:play movies

Type the following command into the query editor:

:play movies



• Change node colors
• Change which node property is displayed
• Run queries with CMD (CTRL) + Enter
• Insert new line with SHIFT + Enter
• Expand the query bar with ESC
• Double-click a node and see what happens!
• Use :clear to clear past results
• CMD (CTRL) + Up Arrow to scroll through past queries

Neo4j Browser 101



The Syntax



Cypher is just ASCII Art



Nodes

Nodes are drawn with parentheses.

()



Relationships

Relationships are drawn as arrows, with additional detail in brackets.

-->

-[:DIRECTED]->



Patterns

Patterns are drawn by connecting nodes and relationships with hyphens, 

optionally specifying a direction with > and < signs.

()-[]-()

()-[]->()

()<-[]-()



The components of a Cypher query

MATCH (m:Movie)
RETURN m

MATCH and RETURN are Cypher keywords
m is a variable
:Movie is a node label



The components of a Cypher query

MATCH (p:Person)-[r:ACTED_IN]->(m:Movie)
RETURN p, r, m

MATCH and RETURN are Cypher keywords
p,r, and m are variables
:Movie is a node label
:ACTED_IN is a relationship type



The components of a Cypher query

MATCH path = (:Person)-[:ACTED_IN]->(:Movie)
RETURN path

MATCH and RETURN are Cypher keywords
path is a variable
:Movie is a node label
:ACTED_IN is a relationship type



MATCH (m:Movie)
RETURN m

Graph versus Tabular results



MATCH (m:Movie)
RETURN m.title, m.released

Properties are accessed with {variable}.{property_key}

Graph versus Tabular results



Case sensitive

Node labels

Relationship types

Property keys

Case insensitive

Cypher keywords

Case sensitivity



Case sensitive

:Person

:ACTED_IN

name

Case insensitive

MaTcH

return

Case sensitivity



The Basics



Cypher: The Basics

Type the following command into the query editor:

:play http://guides.neo4j.com/fundamentals/cypher_the_basics.html 

http://guides.neo4j.com/fundamentals/cypher_the_basics.html


Follow the guides in your browser until you see...



Aggregates



Aggregates

Aggregate queries in Cypher are a little bit different than in SQL as we 

don’t need to specify a grouping key.



Aggregates

We implicitly group by any non aggregate fields in the RETURN 

statement.

// implicitly groups by p.name
MATCH (p:Person)-[:ACTED_IN]->(m:Movie)
RETURN p.name, count(*) AS numberOfMovies



Other aggregate functions

There are other aggregate functions as well. 

You can see the full list on the Cypher refcard:
neo4j.com/docs/cypher-refcard/current/ 

https://neo4j.com/docs/cypher-refcard/current/
https://neo4j.com/docs/cypher-refcard/current/


Continue with the guide

Continue with the guide in your browser



Constraints and Indexes



:schema

Type the following command into the query editor:

:schema



Unique Constraints

We create unique constraints to: 

• ensure uniqueness 

• allow fast lookup of nodes which match label-property pairs.



We create unique constraints to: 

• ensure uniqueness

• allow fast lookup of nodes which match label-property pairs.

CREATE CONSTRAINT ON (label:Label)

ASSERT label.property IS UNIQUE

Unique Constraints



There are three types of unique constraints:

• Unique node property constraint

Unique Constraints



There are three types of unique constraints:

• Unique node property constraint

CREATE CONSTRAINT ON (label:Label)

ASSERT label.property IS UNIQUE

Unique Constraints



There are three types of unique constraints:

• Unique node property constraint

• Node property existence constraint

Unique Constraints



There are three types of unique constraints:

• Unique node property constraint

• Node property existence constraint

CREATE CONSTRAINT ON (label:Label)

ASSERT EXISTS(label.name)

Unique Constraints



There are three types of unique constraints:

• Unique node property constraint

• Node property existence constraint

• Relationship property existence constraint

Unique Constraints



There are three types of unique constraints:

• Unique node property constraint

• Node property existence constraint

• Relationship property existence constraint

CREATE CONSTRAINT ON ()-[rel:REL_TYPE]->()

ASSERT EXISTS(rel.name)

Unique Constraints



Indexes

We create indexes to: 

• allow fast lookup of nodes which match label-property pairs.



Indexes

We create indexes to: 

• allow fast lookup of nodes which match label-property pairs.

CREATE INDEX ON :Label(property)



What are these fast lookups?

The following predicates use indexes:

• Equality
• STARTS WITH

• CONTAINS

• ENDS WITH

• Range searches

• (Non-) existence checks



How are indexes used in neo4j?

Indexes are only used to find the starting points for queries. 

Use index scans to look up rows in 
tables and join them with rows 
from other tables

Use indexes to find the starting 
points for a query. 

Relational Graph



Cypher: The Basics

Type the following command into the query editor:

:play http://guides.neo4j.com/fundamentals/cypher_constraints.html 

http://guides.neo4j.com/fundamentals/cypher_the_basics.html


The MERGE Clause

MERGE (p:Person {name: 'Tom Hanks', oscar: true})

RETURN p

There is not a :Person node with name:'Tom Hanks' and oscar:true in the 
graph, but there is a :Person node with name:'Tom Hanks'. 

What do you think will happen here?



Write queries



The CREATE Clause

CREATE (m:Movie {title:'Mystic River', released:2003})

RETURN m



The SET Clause

MATCH (m:Movie {title: 'Mystic River'})

SET m.tagline = 'We bury our sins here, Dave. We wash them clean.' 

RETURN m



The CREATE Clause

MATCH (m:Movie {title: 'Mystic River'})

MATCH (p:Person {name: 'Kevin Bacon'})

CREATE (p)-[r:ACTED_IN {roles: ['Sean']}]->(m)

RETURN p, r, m



The MERGE Clause

MERGE (p:Person {name: 'Tom Hanks'})

RETURN p



The MERGE Clause

MERGE (p:Person {name: 'Tom Hanks', oscar: true})

RETURN p



The MERGE Clause

MERGE (p:Person {name: 'Tom Hanks'})

SET p.oscar = true

RETURN p



The MERGE Clause

MERGE (p:Person {name: 'Tom Hanks'})-[:ACTED_IN]

       ->(m:Movie {title: 'The Terminal'})

RETURN p, m



The MERGE Clause

MERGE (p:Person {name: 'Tom Hanks'})-[:ACTED_IN]

       ->(m:Movie {title: 'The Terminal'})

RETURN p, m

There is not a :Movie node with title:"The Terminal" in the graph, but 
there is a :Person node with name:"Tom Hanks". 

What do you think will happen here?



MERGE (p:Person {name: 'Tom Hanks'})

MERGE (m:Movie {title: 'The Terminal'})

MERGE (p)-[r:ACTED_IN]->(m)

RETURN p, r, m

The MERGE Clause



MERGE (p:Person {name: 'Your Name'})

 ON CREATE SET p.created = timestamp(), p.updated = 0

 ON MATCH SET p.updated = p.updated + 1

RETURN p.created, p.updated;

ON CREATE and ON MATCH



Style



Clauses

Cypher clauses should be written in all caps and on their own line.

MATCH ...
WHERE ...
RETURN ...



Functions

Functions should be written in lower camel case.

MATCH path = allShortestPaths() ...
WHERE size(nodes(path)) ... 
RETURN ...



Keywords

Keywords should be written in all caps but don’t need to go on their own 
line.

MATCH ...
RETURN collect(DISTINCT n)



Modeling exercise



Modeling conventions



Labels

Labels are upper camel case.

CREATE(n:Person)
CREATE(n:GraphDatabase)
CREATE(n:VeryDescriptiveLabel)



Relationship Types

Relationships are uppercase with underscores to separate words.

CREATE(n)-[:LOVES]->(m)
CREATE(n)-[:REALLY_LOVES]->(m)
CREATE(n)-[:IS_IN_LOVE_WITH]->(m)



Properties are written in lower camelcase.

CREATE(n)
SET n.name = 'Dave'

CREATE(n)
SET n._firstName = 'Dave'

CREATE(n)
SET n.fullName = 'Dave Gordon'

Properties



Modeling exercise: Movie genres



The age old question: should we model them as properties or as nodes?

Adding movie genres

vs



MATCH (m:Movie {title: 'The Matrix'})
SET m.genre = ['Action', 'Sci-Fi']
RETURN m

Genres as properties



MATCH (m:Movie {title: 'Mystic River'})
SET m.genre = ['Action', 'Mystery']
RETURN m

Genres as properties



Accessing a movie’s genres is quick and easy.

MATCH (m:Movie {title:"The Matrix"})
RETURN m.genre;

The good side of properties



Finding movies that share genres is painful and we have a disconnected 
pattern in the MATCH clause - a sure sign you have a modeling issue.

MATCH (m1:Movie), (m2:Movie)
WHERE any(x IN m1.genre WHERE x IN m2.genre)
AND m1 <> m2
RETURN m1, m2;

The bad side of properties



MATCH (m:Movie {title:"The Matrix"})
MERGE (action:Genre {name:"Action"})
MERGE (scifi:Genre {name:"Sci-Fi"})
MERGE (m)-[:IN_GENRE]->(action)
MERGE (m)-[:IN_GENRE]->(scifi)

Genres as nodes



MATCH (m:Movie {title:"Mystic River"})
MERGE (action:Genre {name:"Action"})
MERGE (mystery:Genre {name:"Mystery"})
MERGE (m)-[:IN_GENRE]->(action)
MERGE (m)-[:IN_GENRE]->(mystery)

Genres as nodes



Finding movies that share genres is a natural graph pattern.

MATCH (m1:Movie)-[:IN_GENRE]->(g:Genre),
      (m2:Movie)-[:IN_GENRE]->(g)
RETURN m1, m2, g

The good side of nodes



Accessing the genres of movies requires a bit more typing.

MATCH (m:Movie {title:"The Matrix"}),
      (m)-[:IN_GENRE]->(g:Genre)
RETURN g.name;

The (not too) bad side of nodes



End of Module
Introduction to Cypher

Questions?


